﻿package  {
	import flash.display.InteractiveObject;
	import flash.events.*;
	
	public class IrudiaHitza extends Jokua{
		
		private var hitzZerrenda:Zerrenda;
		private var klik:int =0; //lehenego klik-a den edo bigarrena den esango digu.
		private var aukera1:Object; //lehenengo klik-a egiten dugun objetu
		
		public function IrudiaHitza() {
			
			izena="IrudiaHitza";
			haur = false;
			hitzakAukeratu();
			irudiakPantailaratu();
			hitzakPantailaratu();
		}
		
		private function hitzakAukeratu(){ //pantailaratuko diren hitzen zerrenda aukeratu
			
			var hitzDenak: HitzZerrenda = HitzZerrenda.getInstantzia();
			var i : int;
			
			hitzDenak.nahastu();		
			hitzZerrenda = new Zerrenda();
			for (i=0; i<5 ; i++){
				hitzZerrenda.setElementua(i, hitzDenak.getElementua(i));		
			}
		}//HitzakAukeratu
		
		private function irudiakPantailaratu(){ 
			
			var irudia:Irudia;
			var i:int;
			
			for (i=0; i<hitzZerrenda.luzera(); i++){
				irudia = new Irudia("irudiak/jokuIrudiak/ertainak/" + hitzZerrenda.getElementua(i) + ".png");
				irudia.setIzena("i" + hitzZerrenda.getElementua(i));
				if(i<3){
					irudia.x=600;
					irudia.y=55 + (i * 210);					
				}else{
					irudia.x=800;
					irudia.y=175 + ((i-3) * 210);					
				}
				
				irudia.addEventListener(MouseEvent.CLICK, aukeratu);
				addChild(irudia);
			}			
			
		}//irudiakPantailaratu
		
		private function hitzakPantailaratu(){
			
			var h:Hitza;
			var i : int;
			
			hitzZerrenda.nahastu();
			
			for (i=0; i<hitzZerrenda.luzera(); i++){				
				h = new Hitza("l",hitzZerrenda.getElementua(i));
				h.x=50;
				h.y=75 + (i * 120);
				h.pantailaratu("l");
				h.listenerraGehitu(MouseEvent.CLICK, aukeratu);
				addChild(h);						
			}		
		}
		
		private function aukeratu(eventObject:MouseEvent){
			
			var i:int;
			var objetua:Object;
			
			for (i=0; i<numChildren; i++){			
				
				if(this.getChildAt(i).name==eventObject.target.name){
					objetua = this.getChildAt(i);
				}
			}
		
			if (klik == 0){//lehen aldiz aukeratu dugu				
				aukera1=objetua;
				aukera1.argitu();
				klik++;								
			}else{//bigarren aldiz aukeratu dugu				
				if (erantzunaInterpretatu(aukera1.name,objetua.name)){					
					aukera1.ezabatu(aukeratu);
					objetua.ezabatu(aukeratu);
				}
				klik=0;
			}
			
		}//aukeratu

		private function erantzunaInterpretatu(eran1:String, eran2:String):Boolean
		{							 
			
  			if (eran1 == eran2)
			{//desaktibatu berdinean klik egiteagaitik				
				aukera1.amatatu();
				return (false);
			}
			else if ((eran1==eran2.substring(1,eran2.length))||(eran2==eran1.substring(1,eran1.length)))
			{//bikotea dira
				zuzenak++;
				if (zuzenak == 5){
					irabaziMezuaErakutsi();
				}
				return (true);
			}else{//ez dira bikotea
				aukera1.amatatu();
				return (false);
			}

		}//erantzunaInterpretatu
		
		

	}
}
	

